// File: @openzeppelin/contracts-upgradeable/access/OwnableUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/ContextUpgradeable.sol";
import "../proxy/utils/Initializable.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract OwnableUpgradeable is Initializable, ContextUpgradeable {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    function __Ownable_init() internal onlyInitializing {
        __Ownable_init_unchained();
    }

    function __Ownable_init_unchained() internal onlyInitializing {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }

    /**
     * @dev This empty reserved space is put in place to allow future versions to add new
     * variables without shifting down storage in the inheritance chain.
     * See https://docs.openzeppelin.com/contracts/4.x/upgradeable#storage_gaps
     */
    uint256[49] private __gap;
}

// File: @openzeppelin/contracts-upgradeable/proxy/utils/Initializable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (proxy/utils/Initializable.sol)

pragma solidity ^0.8.2;

import "../../utils/AddressUpgradeable.sol";

/**
 * @dev This is a base contract to aid in writing upgradeable contracts, or any kind of contract that will be deployed
 * behind a proxy. Since proxied contracts do not make use of a constructor, it's common to move constructor logic to an
 * external initializer function, usually called `initialize`. It then becomes necessary to protect this initializer
 * function so it can only be called once. The {initializer} modifier provided by this contract will have this effect.
 *
 * The initialization functions use a version number. Once a version number is used, it is consumed and cannot be
 * reused. This mechanism prevents re-execution of each "step" but allows the creation of new initialization steps in
 * case an upgrade adds a module that needs to be initialized.
 *
 * For example:
 *
 * [.hljs-theme-light.nopadding]
 * ```
 * contract MyToken is ERC20Upgradeable {
 *     function initialize() initializer public {
 *         __ERC20_init("MyToken", "MTK");
 *     }
 * }
 * contract MyTokenV2 is MyToken, ERC20PermitUpgradeable {
 *     function initializeV2() reinitializer(2) public {
 *         __ERC20Permit_init("MyToken");
 *     }
 * }
 * ```
 *
 * TIP: To avoid leaving the proxy in an uninitialized state, the initializer function should be called as early as
 * possible by providing the encoded function call as the `_data` argument to {ERC1967Proxy-constructor}.
 *
 * CAUTION: When used with inheritance, manual care must be taken to not invoke a parent initializer twice, or to ensure
 * that all initializers are idempotent. This is not verified automatically as constructors are by Solidity.
 *
 * [CAUTION]
 * ====
 * Avoid leaving a contract uninitialized.
 *
 * An uninitialized contract can be taken over by an attacker. This applies to both a proxy and its implementation
 * contract, which may impact the proxy. To prevent the implementation contract from being used, you should invoke
 * the {_disableInitializers} function in the constructor to automatically lock it when it is deployed:
 *
 * [.hljs-theme-light.nopadding]
 * ```
 * /// @custom:oz-upgrades-unsafe-allow constructor
 * constructor() {
 *     _disableInitializers();
 * }
 * ```
 * ====
 */
abstract contract Initializable {
    /**
     * @dev Indicates that the contract has been initialized.
     * @custom:oz-retyped-from bool
     */
    uint8 private _initialized;

    /**
     * @dev Indicates that the contract is in the process of being initialized.
     */
    bool private _initializing;

    /**
     * @dev Triggered when the contract has been initialized or reinitialized.
     */
    event Initialized(uint8 version);

    /**
     * @dev A modifier that defines a protected initializer function that can be invoked at most once. In its scope,
     * `onlyInitializing` functions can be used to initialize parent contracts. Equivalent to `reinitializer(1)`.
     */
    modifier initializer() {
        bool isTopLevelCall = !_initializing;
        require(
            (isTopLevelCall && _initialized < 1) || (!AddressUpgradeable.isContract(address(this)) && _initialized == 1),
            "Initializable: contract is already initialized"
        );
        _initialized = 1;
        if (isTopLevelCall) {
            _initializing = true;
        }
        _;
        if (isTopLevelCall) {
            _initializing = false;
            emit Initialized(1);
        }
    }

    /**
     * @dev A modifier that defines a protected reinitializer function that can be invoked at most once, and only if the
     * contract hasn't been initialized to a greater version before. In its scope, `onlyInitializing` functions can be
     * used to initialize parent contracts.
     *
     * `initializer` is equivalent to `reinitializer(1)`, so a reinitializer may be used after the original
     * initialization step. This is essential to configure modules that are added through upgrades and that require
     * initialization.
     *
     * Note that versions can jump in increments greater than 1; this implies that if multiple reinitializers coexist in
     * a contract, executing them in the right order is up to the developer or operator.
     */
    modifier reinitializer(uint8 version) {
        require(!_initializing && _initialized < version, "Initializable: contract is already initialized");
        _initialized = version;
        _initializing = true;
        _;
        _initializing = false;
        emit Initialized(version);
    }

    /**
     * @dev Modifier to protect an initialization function so that it can only be invoked by functions with the
     * {initializer} and {reinitializer} modifiers, directly or indirectly.
     */
    modifier onlyInitializing() {
        require(_initializing, "Initializable: contract is not initializing");
        _;
    }

    /**
     * @dev Locks the contract, preventing any future reinitialization. This cannot be part of an initializer call.
     * Calling this in the constructor of a contract will prevent that contract from being initialized or reinitialized
     * to any version. It is recommended to use this to lock implementation contracts that are designed to be called
     * through proxies.
     */
    function _disableInitializers() internal virtual {
        require(!_initializing, "Initializable: contract is initializing");
        if (_initialized < type(uint8).max) {
            _initialized = type(uint8).max;
            emit Initialized(type(uint8).max);
        }
    }
}

// File: @openzeppelin/contracts-upgradeable/token/ERC721/ERC721Upgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721Upgradeable.sol";
import "./IERC721ReceiverUpgradeable.sol";
import "./extensions/IERC721MetadataUpgradeable.sol";
import "../../utils/AddressUpgradeable.sol";
import "../../utils/ContextUpgradeable.sol";
import "../../utils/StringsUpgradeable.sol";
import "../../utils/introspection/ERC165Upgradeable.sol";
import "../../proxy/utils/Initializable.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721Upgradeable is Initializable, ContextUpgradeable, ERC165Upgradeable, IERC721Upgradeable, IERC721MetadataUpgradeable {
    using AddressUpgradeable for address;
    using StringsUpgradeable for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    function __ERC721_init(string memory name_, string memory symbol_) internal onlyInitializing {
        __ERC721_init_unchained(name_, symbol_);
    }

    function __ERC721_init_unchained(string memory name_, string memory symbol_) internal onlyInitializing {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165Upgradeable, IERC165Upgradeable) returns (bool) {
        return
            interfaceId == type(IERC721Upgradeable).interfaceId ||
            interfaceId == type(IERC721MetadataUpgradeable).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721Upgradeable.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner nor approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721Upgradeable.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721Upgradeable.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721Upgradeable.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721Upgradeable.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721ReceiverUpgradeable(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721ReceiverUpgradeable.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev This empty reserved space is put in place to allow future versions to add new
     * variables without shifting down storage in the inheritance chain.
     * See https://docs.openzeppelin.com/contracts/4.x/upgradeable#storage_gaps
     */
    uint256[44] private __gap;
}

// File: @openzeppelin/contracts-upgradeable/token/ERC721/extensions/IERC721MetadataUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721Upgradeable.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721MetadataUpgradeable is IERC721Upgradeable {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts-upgradeable/token/ERC721/IERC721ReceiverUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721ReceiverUpgradeable {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts-upgradeable/token/ERC721/IERC721Upgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165Upgradeable.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721Upgradeable is IERC165Upgradeable {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts-upgradeable/utils/AddressUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library AddressUpgradeable {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts-upgradeable/utils/ContextUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;
import "../proxy/utils/Initializable.sol";

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract ContextUpgradeable is Initializable {
    function __Context_init() internal onlyInitializing {
    }

    function __Context_init_unchained() internal onlyInitializing {
    }
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }

    /**
     * @dev This empty reserved space is put in place to allow future versions to add new
     * variables without shifting down storage in the inheritance chain.
     * See https://docs.openzeppelin.com/contracts/4.x/upgradeable#storage_gaps
     */
    uint256[50] private __gap;
}

// File: @openzeppelin/contracts-upgradeable/utils/introspection/ERC165Upgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165Upgradeable.sol";
import "../../proxy/utils/Initializable.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165Upgradeable is Initializable, IERC165Upgradeable {
    function __ERC165_init() internal onlyInitializing {
    }

    function __ERC165_init_unchained() internal onlyInitializing {
    }
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165Upgradeable).interfaceId;
    }

    /**
     * @dev This empty reserved space is put in place to allow future versions to add new
     * variables without shifting down storage in the inheritance chain.
     * See https://docs.openzeppelin.com/contracts/4.x/upgradeable#storage_gaps
     */
    uint256[50] private __gap;
}

// File: @openzeppelin/contracts-upgradeable/utils/introspection/IERC165Upgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165Upgradeable {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts-upgradeable/utils/StringsUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library StringsUpgradeable {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/interfaces/draft-IERC1822.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (interfaces/draft-IERC1822.sol)

pragma solidity ^0.8.0;

/**
 * @dev ERC1822: Universal Upgradeable Proxy Standard (UUPS) documents a method for upgradeability through a simplified
 * proxy whose upgrades are fully controlled by the current implementation.
 */
interface IERC1822Proxiable {
    /**
     * @dev Returns the storage slot that the proxiable contract assumes is being used to store the implementation
     * address.
     *
     * IMPORTANT: A proxy pointing at a proxiable contract should not be considered proxiable itself, because this risks
     * bricking a proxy that upgrades to it, by delegating to itself until out of gas. Thus it is critical that this
     * function revert if invoked through a proxy.
     */
    function proxiableUUID() external view returns (bytes32);
}

// File: @openzeppelin/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: @openzeppelin/contracts/proxy/beacon/IBeacon.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (proxy/beacon/IBeacon.sol)

pragma solidity ^0.8.0;

/**
 * @dev This is the interface that {BeaconProxy} expects of its beacon.
 */
interface IBeacon {
    /**
     * @dev Must return an address that can be used as a delegate call target.
     *
     * {BeaconProxy} will check that this address is a contract.
     */
    function implementation() external view returns (address);
}

// File: @openzeppelin/contracts/proxy/ERC1967/ERC1967Proxy.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (proxy/ERC1967/ERC1967Proxy.sol)

pragma solidity ^0.8.0;

import "../Proxy.sol";
import "./ERC1967Upgrade.sol";

/**
 * @dev This contract implements an upgradeable proxy. It is upgradeable because calls are delegated to an
 * implementation address that can be changed. This address is stored in storage in the location specified by
 * https://eips.ethereum.org/EIPS/eip-1967[EIP1967], so that it doesn't conflict with the storage layout of the
 * implementation behind the proxy.
 */
contract ERC1967Proxy is Proxy, ERC1967Upgrade {
    /**
     * @dev Initializes the upgradeable proxy with an initial implementation specified by `_logic`.
     *
     * If `_data` is nonempty, it's used as data in a delegate call to `_logic`. This will typically be an encoded
     * function call, and allows initializing the storage of the proxy like a Solidity constructor.
     */
    constructor(address _logic, bytes memory _data) payable {
        _upgradeToAndCall(_logic, _data, false);
    }

    /**
     * @dev Returns the current implementation address.
     */
    function _implementation() internal view virtual override returns (address impl) {
        return ERC1967Upgrade._getImplementation();
    }
}

// File: @openzeppelin/contracts/proxy/ERC1967/ERC1967Upgrade.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (proxy/ERC1967/ERC1967Upgrade.sol)

pragma solidity ^0.8.2;

import "../beacon/IBeacon.sol";
import "../../interfaces/draft-IERC1822.sol";
import "../../utils/Address.sol";
import "../../utils/StorageSlot.sol";

/**
 * @dev This abstract contract provides getters and event emitting update functions for
 * https://eips.ethereum.org/EIPS/eip-1967[EIP1967] slots.
 *
 * _Available since v4.1._
 *
 * @custom:oz-upgrades-unsafe-allow delegatecall
 */
abstract contract ERC1967Upgrade {
    // This is the keccak-256 hash of "eip1967.proxy.rollback" subtracted by 1
    bytes32 private constant _ROLLBACK_SLOT = 0x4910fdfa16fed3260ed0e7147f7cc6da11a60208b5b9406d12a635614ffd9143;

    /**
     * @dev Storage slot with the address of the current implementation.
     * This is the keccak-256 hash of "eip1967.proxy.implementation" subtracted by 1, and is
     * validated in the constructor.
     */
    bytes32 internal constant _IMPLEMENTATION_SLOT = 0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc;

    /**
     * @dev Emitted when the implementation is upgraded.
     */
    event Upgraded(address indexed implementation);

    /**
     * @dev Returns the current implementation address.
     */
    function _getImplementation() internal view returns (address) {
        return StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value;
    }

    /**
     * @dev Stores a new address in the EIP1967 implementation slot.
     */
    function _setImplementation(address newImplementation) private {
        require(Address.isContract(newImplementation), "ERC1967: new implementation is not a contract");
        StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value = newImplementation;
    }

    /**
     * @dev Perform implementation upgrade
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeTo(address newImplementation) internal {
        _setImplementation(newImplementation);
        emit Upgraded(newImplementation);
    }

    /**
     * @dev Perform implementation upgrade with additional setup call.
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeToAndCall(
        address newImplementation,
        bytes memory data,
        bool forceCall
    ) internal {
        _upgradeTo(newImplementation);
        if (data.length > 0 || forceCall) {
            Address.functionDelegateCall(newImplementation, data);
        }
    }

    /**
     * @dev Perform implementation upgrade with security checks for UUPS proxies, and additional setup call.
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeToAndCallUUPS(
        address newImplementation,
        bytes memory data,
        bool forceCall
    ) internal {
        // Upgrades from old implementations will perform a rollback test. This test requires the new
        // implementation to upgrade back to the old, non-ERC1822 compliant, implementation. Removing
        // this special case will break upgrade paths from old UUPS implementation to new ones.
        if (StorageSlot.getBooleanSlot(_ROLLBACK_SLOT).value) {
            _setImplementation(newImplementation);
        } else {
            try IERC1822Proxiable(newImplementation).proxiableUUID() returns (bytes32 slot) {
                require(slot == _IMPLEMENTATION_SLOT, "ERC1967Upgrade: unsupported proxiableUUID");
            } catch {
                revert("ERC1967Upgrade: new implementation is not UUPS");
            }
            _upgradeToAndCall(newImplementation, data, forceCall);
        }
    }

    /**
     * @dev Storage slot with the admin of the contract.
     * This is the keccak-256 hash of "eip1967.proxy.admin" subtracted by 1, and is
     * validated in the constructor.
     */
    bytes32 internal constant _ADMIN_SLOT = 0xb53127684a568b3173ae13b9f8a6016e243e63b6e8ee1178d6a717850b5d6103;

    /**
     * @dev Emitted when the admin account has changed.
     */
    event AdminChanged(address previousAdmin, address newAdmin);

    /**
     * @dev Returns the current admin.
     */
    function _getAdmin() internal view returns (address) {
        return StorageSlot.getAddressSlot(_ADMIN_SLOT).value;
    }

    /**
     * @dev Stores a new address in the EIP1967 admin slot.
     */
    function _setAdmin(address newAdmin) private {
        require(newAdmin != address(0), "ERC1967: new admin is the zero address");
        StorageSlot.getAddressSlot(_ADMIN_SLOT).value = newAdmin;
    }

    /**
     * @dev Changes the admin of the proxy.
     *
     * Emits an {AdminChanged} event.
     */
    function _changeAdmin(address newAdmin) internal {
        emit AdminChanged(_getAdmin(), newAdmin);
        _setAdmin(newAdmin);
    }

    /**
     * @dev The storage slot of the UpgradeableBeacon contract which defines the implementation for this proxy.
     * This is bytes32(uint256(keccak256('eip1967.proxy.beacon')) - 1)) and is validated in the constructor.
     */
    bytes32 internal constant _BEACON_SLOT = 0xa3f0ad74e5423aebfd80d3ef4346578335a9a72aeaee59ff6cb3582b35133d50;

    /**
     * @dev Emitted when the beacon is upgraded.
     */
    event BeaconUpgraded(address indexed beacon);

    /**
     * @dev Returns the current beacon.
     */
    function _getBeacon() internal view returns (address) {
        return StorageSlot.getAddressSlot(_BEACON_SLOT).value;
    }

    /**
     * @dev Stores a new beacon in the EIP1967 beacon slot.
     */
    function _setBeacon(address newBeacon) private {
        require(Address.isContract(newBeacon), "ERC1967: new beacon is not a contract");
        require(
            Address.isContract(IBeacon(newBeacon).implementation()),
            "ERC1967: beacon implementation is not a contract"
        );
        StorageSlot.getAddressSlot(_BEACON_SLOT).value = newBeacon;
    }

    /**
     * @dev Perform beacon upgrade with additional setup call. Note: This upgrades the address of the beacon, it does
     * not upgrade the implementation contained in the beacon (see {UpgradeableBeacon-_setImplementation} for that).
     *
     * Emits a {BeaconUpgraded} event.
     */
    function _upgradeBeaconToAndCall(
        address newBeacon,
        bytes memory data,
        bool forceCall
    ) internal {
        _setBeacon(newBeacon);
        emit BeaconUpgraded(newBeacon);
        if (data.length > 0 || forceCall) {
            Address.functionDelegateCall(IBeacon(newBeacon).implementation(), data);
        }
    }
}

// File: @openzeppelin/contracts/proxy/Proxy.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (proxy/Proxy.sol)

pragma solidity ^0.8.0;

/**
 * @dev This abstract contract provides a fallback function that delegates all calls to another contract using the EVM
 * instruction `delegatecall`. We refer to the second contract as the _implementation_ behind the proxy, and it has to
 * be specified by overriding the virtual {_implementation} function.
 *
 * Additionally, delegation to the implementation can be triggered manually through the {_fallback} function, or to a
 * different contract through the {_delegate} function.
 *
 * The success and return data of the delegated call will be returned back to the caller of the proxy.
 */
abstract contract Proxy {
    /**
     * @dev Delegates the current call to `implementation`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _delegate(address implementation) internal virtual {
        assembly {
            // Copy msg.data. We take full control of memory in this inline assembly
            // block because it will not return to Solidity code. We overwrite the
            // Solidity scratch pad at memory position 0.
            calldatacopy(0, 0, calldatasize())

            // Call the implementation.
            // out and outsize are 0 because we don't know the size yet.
            let result := delegatecall(gas(), implementation, 0, calldatasize(), 0, 0)

            // Copy the returned data.
            returndatacopy(0, 0, returndatasize())

            switch result
            // delegatecall returns 0 on error.
            case 0 {
                revert(0, returndatasize())
            }
            default {
                return(0, returndatasize())
            }
        }
    }

    /**
     * @dev This is a virtual function that should be overridden so it returns the address to which the fallback function
     * and {_fallback} should delegate.
     */
    function _implementation() internal view virtual returns (address);

    /**
     * @dev Delegates the current call to the address returned by `_implementation()`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _fallback() internal virtual {
        _beforeFallback();
        _delegate(_implementation());
    }

    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if no other
     * function in the contract matches the call data.
     */
    fallback() external payable virtual {
        _fallback();
    }

    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if call data
     * is empty.
     */
    receive() external payable virtual {
        _fallback();
    }

    /**
     * @dev Hook that is called before falling back to the implementation. Can happen as part of a manual `_fallback`
     * call, or as part of the Solidity `fallback` or `receive` functions.
     *
     * If overridden should call `super._beforeFallback()`.
     */
    function _beforeFallback() internal virtual {}
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Base64.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        /// @solidity memory-safe-assembly
        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Create2.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Create2.sol)

pragma solidity ^0.8.0;

/**
 * @dev Helper to make usage of the `CREATE2` EVM opcode easier and safer.
 * `CREATE2` can be used to compute in advance the address where a smart
 * contract will be deployed, which allows for interesting new mechanisms known
 * as 'counterfactual interactions'.
 *
 * See the https://eips.ethereum.org/EIPS/eip-1014#motivation[EIP] for more
 * information.
 */
library Create2 {
    /**
     * @dev Deploys a contract using `CREATE2`. The address where the contract
     * will be deployed can be known in advance via {computeAddress}.
     *
     * The bytecode for a contract can be obtained from Solidity with
     * `type(contractName).creationCode`.
     *
     * Requirements:
     *
     * - `bytecode` must not be empty.
     * - `salt` must have not been used for `bytecode` already.
     * - the factory must have a balance of at least `amount`.
     * - if `amount` is non-zero, `bytecode` must have a `payable` constructor.
     */
    function deploy(
        uint256 amount,
        bytes32 salt,
        bytes memory bytecode
    ) internal returns (address) {
        address addr;
        require(address(this).balance >= amount, "Create2: insufficient balance");
        require(bytecode.length != 0, "Create2: bytecode length is zero");
        /// @solidity memory-safe-assembly
        assembly {
            addr := create2(amount, add(bytecode, 0x20), mload(bytecode), salt)
        }
        require(addr != address(0), "Create2: Failed on deploy");
        return addr;
    }

    /**
     * @dev Returns the address where a contract will be stored if deployed via {deploy}. Any change in the
     * `bytecodeHash` or `salt` will result in a new destination address.
     */
    function computeAddress(bytes32 salt, bytes32 bytecodeHash) internal view returns (address) {
        return computeAddress(salt, bytecodeHash, address(this));
    }

    /**
     * @dev Returns the address where a contract will be stored if deployed via {deploy} from a contract located at
     * `deployer`. If `deployer` is this contract's address, returns the same value as {computeAddress}.
     */
    function computeAddress(
        bytes32 salt,
        bytes32 bytecodeHash,
        address deployer
    ) internal pure returns (address) {
        bytes32 _data = keccak256(abi.encodePacked(bytes1(0xff), deployer, salt, bytecodeHash));
        return address(uint160(uint256(_data)));
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC1820Implementer.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC1820Implementer.sol)

pragma solidity ^0.8.0;

import "./IERC1820Implementer.sol";

/**
 * @dev Implementation of the {IERC1820Implementer} interface.
 *
 * Contracts may inherit from this and call {_registerInterfaceForAddress} to
 * declare their willingness to be implementers.
 * {IERC1820Registry-setInterfaceImplementer} should then be called for the
 * registration to be complete.
 */
contract ERC1820Implementer is IERC1820Implementer {
    bytes32 private constant _ERC1820_ACCEPT_MAGIC = keccak256("ERC1820_ACCEPT_MAGIC");

    mapping(bytes32 => mapping(address => bool)) private _supportedInterfaces;

    /**
     * @dev See {IERC1820Implementer-canImplementInterfaceForAddress}.
     */
    function canImplementInterfaceForAddress(bytes32 interfaceHash, address account)
        public
        view
        virtual
        override
        returns (bytes32)
    {
        return _supportedInterfaces[interfaceHash][account] ? _ERC1820_ACCEPT_MAGIC : bytes32(0x00);
    }

    /**
     * @dev Declares the contract as willing to be an implementer of
     * `interfaceHash` for `account`.
     *
     * See {IERC1820Registry-setInterfaceImplementer} and
     * {IERC1820Registry-interfaceHash}.
     */
    function _registerInterfaceForAddress(bytes32 interfaceHash, address account) internal virtual {
        _supportedInterfaces[interfaceHash][account] = true;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC1820Implementer.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC1820Implementer.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface for an ERC1820 implementer, as defined in the
 * https://eips.ethereum.org/EIPS/eip-1820#interface-implementation-erc1820implementerinterface[EIP].
 * Used by contracts that will be registered as implementers in the
 * {IERC1820Registry}.
 */
interface IERC1820Implementer {
    /**
     * @dev Returns a special value (`ERC1820_ACCEPT_MAGIC`) if this contract
     * implements `interfaceHash` for `account`.
     *
     * See {IERC1820Registry-setInterfaceImplementer}.
     */
    function canImplementInterfaceForAddress(bytes32 interfaceHash, address account) external view returns (bytes32);
}

// File: @openzeppelin/contracts/utils/StorageSlot.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/StorageSlot.sol)

pragma solidity ^0.8.0;

/**
 * @dev Library for reading and writing primitive types to specific storage slots.
 *
 * Storage slots are often used to avoid storage conflict when dealing with upgradeable contracts.
 * This library helps with reading and writing to such slots without the need for inline assembly.
 *
 * The functions in this library return Slot structs that contain a `value` member that can be used to read or write.
 *
 * Example usage to set ERC1967 implementation slot:
 * ```
 * contract ERC1967 {
 *     bytes32 internal constant _IMPLEMENTATION_SLOT = 0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc;
 *
 *     function _getImplementation() internal view returns (address) {
 *         return StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value;
 *     }
 *
 *     function _setImplementation(address newImplementation) internal {
 *         require(Address.isContract(newImplementation), "ERC1967: new implementation is not a contract");
 *         StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value = newImplementation;
 *     }
 * }
 * ```
 *
 * _Available since v4.1 for `address`, `bool`, `bytes32`, and `uint256`._
 */
library StorageSlot {
    struct AddressSlot {
        address value;
    }

    struct BooleanSlot {
        bool value;
    }

    struct Bytes32Slot {
        bytes32 value;
    }

    struct Uint256Slot {
        uint256 value;
    }

    /**
     * @dev Returns an `AddressSlot` with member `value` located at `slot`.
     */
    function getAddressSlot(bytes32 slot) internal pure returns (AddressSlot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }

    /**
     * @dev Returns an `BooleanSlot` with member `value` located at `slot`.
     */
    function getBooleanSlot(bytes32 slot) internal pure returns (BooleanSlot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }

    /**
     * @dev Returns an `Bytes32Slot` with member `value` located at `slot`.
     */
    function getBytes32Slot(bytes32 slot) internal pure returns (Bytes32Slot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }

    /**
     * @dev Returns an `Uint256Slot` with member `value` located at `slot`.
     */
    function getUint256Slot(bytes32 slot) internal pure returns (Uint256Slot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: solmate/src/utils/Bytes32AddressLib.sol



// SPDX-License-Identifier: AGPL-3.0-only
pragma solidity >=0.8.0;

/// @notice Library for converting between addresses and bytes32 values.
/// @author Solmate (https://github.com/transmissions11/solmate/blob/main/src/utils/Bytes32AddressLib.sol)
library Bytes32AddressLib {
    function fromLast20Bytes(bytes32 bytesValue) internal pure returns (address) {
        return address(uint160(uint256(bytesValue)));
    }

    function fillLast12Bytes(address addressValue) internal pure returns (bytes32) {
        return bytes32(bytes20(addressValue));
    }
}

// File: solmate/src/utils/CREATE3.sol



// SPDX-License-Identifier: AGPL-3.0-only
pragma solidity >=0.8.0;

import {Bytes32AddressLib} from "./Bytes32AddressLib.sol";

/// @notice Deploy to deterministic addresses without an initcode factor.
/// @author Solmate (https://github.com/transmissions11/solmate/blob/main/src/utils/CREATE3.sol)
/// @author Modified from 0xSequence (https://github.com/0xSequence/create3/blob/master/contracts/Create3.sol)
library CREATE3 {
    using Bytes32AddressLib for bytes32;

    //--------------------------------------------------------------------------------//
    // Opcode     | Opcode + Arguments    | Description      | Stack View             //
    //--------------------------------------------------------------------------------//
    // 0x36       |  0x36                 | CALLDATASIZE     | size                   //
    // 0x3d       |  0x3d                 | RETURNDATASIZE   | 0 size                 //
    // 0x3d       |  0x3d                 | RETURNDATASIZE   | 0 0 size               //
    // 0x37       |  0x37                 | CALLDATACOPY     |                        //
    // 0x36       |  0x36                 | CALLDATASIZE     | size                   //
    // 0x3d       |  0x3d                 | RETURNDATASIZE   | 0 size                 //
    // 0x34       |  0x34                 | CALLVALUE        | value 0 size           //
    // 0xf0       |  0xf0                 | CREATE           | newContract            //
    //--------------------------------------------------------------------------------//
    // Opcode     | Opcode + Arguments    | Description      | Stack View             //
    //--------------------------------------------------------------------------------//
    // 0x67       |  0x67XXXXXXXXXXXXXXXX | PUSH8 bytecode   | bytecode               //
    // 0x3d       |  0x3d                 | RETURNDATASIZE   | 0 bytecode             //
    // 0x52       |  0x52                 | MSTORE           |                        //
    // 0x60       |  0x6008               | PUSH1 08         | 8                      //
    // 0x60       |  0x6018               | PUSH1 18         | 24 8                   //
    // 0xf3       |  0xf3                 | RETURN           |                        //
    //--------------------------------------------------------------------------------//
    bytes internal constant PROXY_BYTECODE = hex"67_36_3d_3d_37_36_3d_34_f0_3d_52_60_08_60_18_f3";

    bytes32 internal constant PROXY_BYTECODE_HASH = keccak256(PROXY_BYTECODE);

    function deploy(
        bytes32 salt,
        bytes memory creationCode,
        uint256 value
    ) internal returns (address deployed) {
        bytes memory proxyChildBytecode = PROXY_BYTECODE;

        address proxy;
        /// @solidity memory-safe-assembly
        assembly {
            // Deploy a new contract with our pre-made bytecode via CREATE2.
            // We start 32 bytes into the code to avoid copying the byte length.
            proxy := create2(0, add(proxyChildBytecode, 32), mload(proxyChildBytecode), salt)
        }
        require(proxy != address(0), "DEPLOYMENT_FAILED");

        deployed = getDeployed(salt);
        (bool success, ) = proxy.call{value: value}(creationCode);
        require(success && deployed.code.length != 0, "INITIALIZATION_FAILED");
    }

    function getDeployed(bytes32 salt) internal view returns (address) {
        address proxy = keccak256(
            abi.encodePacked(
                // Prefix:
                bytes1(0xFF),
                // Creator:
                address(this),
                // Salt:
                salt,
                // Bytecode hash:
                PROXY_BYTECODE_HASH
            )
        ).fromLast20Bytes();

        return
            keccak256(
                abi.encodePacked(
                    // 0xd6 = 0xc0 (short RLP prefix) + 0x16 (length of: 0x94 ++ proxy ++ 0x01)
                    // 0x94 = 0x80 + 0x14 (0x14 = the length of an address, 20 bytes, in hex)
                    hex"d6_94",
                    proxy,
                    hex"01" // Nonce of the proxy contract (1)
                )
            ).fromLast20Bytes();
    }
}

// File: src/contracts/Errors.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

///@title Errors
///@notice Defines the different errors of Gamma Address contracts
library Errors {
  /// @notice Function was not called by owner.
  string public constant OWNABLE_NOT_OWNER = "OWNABLE_NOT_OWNER";
  /// @notice Function was not called by the NFT address owner.
  string public constant ERC721_NOT_TOKEN_OWNER = "ERC721_NOT_TOKEN_OWNER";
  /// @notice Factory can't be used to mint an NFT address.
  string public constant UNAUTHORIZED_FACTORY = "UNAUTHORIZED_FACTORY";
  /// @notice ETH value insufficient to mint an NFT address. See `Registrar._mintPrice` value.
  string public constant INSUFFICIENT_MINT_PAYMENT = "INSUFFICIENT_MINT_PAYMENT";
  /// @notice Contract's balance is less than the requested amount to withdraw.
  string public constant INSUFFICIENT_FUNDS = "INSUFFICIENT_FUNDS";
  /// @notice Transfer failed.
  string public constant TRANSFER_FAILED = "TRANSFER_FAILED";
  /// @notice The NFT address can't be minted because it already exists.
  string public constant ADDRESS_ALREADY_MINTED = "ADDRESS_ALREADY_MINTED";
  /// @notice Function was not called by the Registrar.
  string public constant NOT_REGISTRAR = "NOT_REGISTRAR";
  /// @notice Deployment problem. The smart contract was not deployed to the correct address.
  string public constant INVALID_ADDRESS_CREATED = "INVALID_ADDRESS_CREATED";
  /// @notice Royalties should be 100%(10000 basis points) or less.
  string public constant INVALID_ROYALTIES = "INVALID_ROYALTIES";
  /// @notice Sender doesn't have deploy permission
  string public constant NO_DEPLOY_PERMISSION = "NO_DEPLOY_PERMISSION";
}

// File: src/contracts/IFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

/// @title IFactory interface
/// @dev Factories are used to register and deploy NFT addresses.
interface IFactory {
  /// @dev Deploys a smart contract to an NFT address
  /// @param nftAddress NFT address
  /// @param creationCode Smart contract bytecode (i.e. bytecode containing the constructor and
  /// initialization data)
  function deploy(address nftAddress, bytes memory creationCode) external;

  /// @dev Mints a new NFT address
  /// @param owner Owner of the NFT address
  /// @param salt Salt used to generate the NFT Address
  function mint(address owner, uint256 salt) external returns (address payable);
}

// File: src/contracts/IMetaData.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

import {Registrar} from "./Registrar.sol";

/// @title IMetaData interface
/// @notice The metadata of NFT addresses is fully generated on-chain, including the SVG image.
/// @dev The metadata logic is encapsulated here so it can be easier to update without doing
/// a full upgrade of the Registrar.
interface IMetaData {
  /// @notice Returns the URI for an NFT address
  /// @param tokenId ID of the NFT address
  /// @param registrar Registrar
  function getTokenURI(uint256 tokenId, Registrar registrar) external view returns (string memory);

  /// @notice Returns the SVG image for an NFT address
  /// @param tokenId ID of the NFT address
  /// @param registrar Registrar
  function getImage(uint256 tokenId, Registrar registrar) external view returns (string memory);

  /// @notice Returns the metadata for and NFT address
  /// @param tokenId ID of the NFT address
  /// @param registrar Registrar
  function getMetaData(uint256 tokenId, Registrar registrar) external view returns (string memory);
}

// File: src/contracts/MetaData.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

import {Base64} from "@openzeppelin/contracts/utils/Base64.sol";
import {IMetaData} from "./IMetaData.sol";
import {Registrar} from "./Registrar.sol";
import {Strings} from "@openzeppelin/contracts/utils/Strings.sol";

/// @title MetaData
/// @notice The metadata of NFT addresses is fully generated on-chain, including the SVG image.
/// @dev The metadata logic is encapsulated here so it can be easier to update without doing
/// a full upgrade of the Registrar.
contract MetaData is IMetaData {
  /// @notice Returns the URI for an NFT address
  /// @param tokenId ID of the NFT address
  /// @param registrar Registrar
  function getTokenURI(uint256 tokenId, Registrar registrar) external pure returns (string memory) {
    return
      string(
        abi.encodePacked(
          "data:application/json;base64,",
          Base64.encode(bytes(getMetaData(tokenId, registrar)))
        )
      );
  }

  /// @notice Returns the SVG image for an NFT address
  /// @param tokenId ID of the NFT address
  /// @param registrar Registrar
  function getImage(uint256 tokenId, Registrar registrar) public pure returns (string memory) {
    return
      string(
        abi.encodePacked(
          '<svg width="424" height="524" viewBox="0 0 452 552" fill="none" xmlns="http://www.w3.org/2000/svg"><g><svg class="nft-address" width="424" height="524" viewBox="0 0 424 524" fill="none" x="14" y="14"><style>                @font-face { font-family: \'Sofia Sans Extra Condensed\'; font-style: normal; src: url(data:application/font-woff2;charset=utf-8;base64,d09GMgABAAAAACDIABAAAAAAbhQAACBoAAEAAAAAAAAAAAAAAAAAAAAAAAAAAAAAP0ZGVE0cGh4b+XocgnQGYACEEhEICsV4tx0LgWIAATYCJAODOgQgBYd8B4NIDAcbvWQF3BhntwNCIX3vJKKYFNH/lwRtjFD4dmjVCodgkoq0pCJk4oK0sA9pwzTVVO9bkxT7ydH482G7u2DsXH3Awj8WLIW0dMWdD/xsBK6w5uHHsOoIjX2SSxSN9W/1HHxiVkAcQFDALhbIkvpx0WwBVfJ0SGaPp6n2s6vdBzLKIksmWUBkkWVgyUgykELkEqTslIBTzPWAkwJQ2rv7gBl0y5LWk/7GkYrL9roXCT1qP22whhVC1QQ51GndO8zrD2PXB6AyBdCQVGbJcmQZdIKjIFWjGiAFasU1hbP+w4+CeChg50HAmngUclIis8kXwD8BNJfNUpIDBNUHW2EfoZQUKXAnXb/AJcogSqjp/amu/+lLlhWUXMRfnHoelkhluj2dOkxf773vZ/0nMF4tya4dtpzwD8u9k3RKrPp/GejkEpFSCBCPBYAJkPZcp0zORrB12HoZxnbHeaJp65rHZALWorXRPu7thlWbGc65+7P7FykiIiVIkBBCCUWK93z3sumDTw3y2iQlFz+IlO2L6I9iCODXvvleAH57+s+zAL+/9dknGGA4MBEoDIEiBgcUEGColF6CXXs6+0Fy0/4da6C6e2bXBlhxgIYClvlz1y07NoCA21o47EqU55SbnDUerARGSgbokailZRXGiLeeBMPehFqMSFvRYtkbZxVpJSuGsllhDB5HpIZRDjvzOg7MWFz3Jbh0eodAQ6izLAE0I716AZFSInYBhouY/H1lsryiaiBIlx8vjlOFrgoYGb1N7dosEnPslPIzFdUUBViRLKGB+84qE70b6hXURnmbf53PmC6+wVVLXK0CCz9r46rGG/SxT+nZXXaj0zo1lcZGcUHWCkdHetikukZFdJJUwsLn4py6CHkpmj5TqUmZcuGuescYSTsXGMIdpUqAaPD5S0drOF47f1A8GlTPJQi74bUMHdPm/FR9OvDB3ZExWfp796oH0L68zViA9zoVgJuyg9aw7BSmUGXprlcYlfcc5ksP3fY1EbaRWSTy1YuEX/zlTzdr4W2gQgs6zoo9ztCuQxoQNGwiXzoisKxQvjbHTGGD5ewrqyZwFXZb4+lFnRc5HpfQjy16FNtrPFjZWSdvdymTkJPNJxG6H++KxftMwW53ngT6+jFdAsH/TgCSQOopUMWEwTug0AO3ZgEXSG3dCqrzVpj1Xa4hYRUcImCc0EdQUyPPtnUlwAFFJU6FKvJiu8KhudMvjMInkiVSK4soubum3a1i7ovcWBhPOrkKfgdA7Dh6LeYFjNWgjKByoeo0nybHsWLCotrXHV+Wuo7anuYvrcvXmXS6XcmYtD7ua53GeALYGsLeUkWD4yeaBs8POuFoa9ZmJU7fuw9XZRR73bVwf8Z5zq62lHWXzXTqI34dIQbAkp3YIn2bkL3rcbsRgQ3ORPVSO4+y72BHdsN9yJE7PFe5uW9M+OwFVlWaMJpiROfZBE9xX/4FLbbyquSE9kWu7Bo9zjMdDG8m53cDlBnsejpM1e8nbWVlBPhBhW8hvMsakmTlGHDhv9ECi9VglkQkUfxpbUo5FTiMALxSZAC/gPZdydNwFZ2uttxaMXi1ITRyA4vkHGxYQaWH3ctzG6NwyxxV+yByB9S26nHczPKYBlBV+Hi9qoyCxigi7SFa0g5qzc+MvqnCBL+KpjNDRQWLyJaoppSIOa5q9h4hpFo2o4ELPirrGS1RkX9jlF+C15f/7czBmB0bPBRjFV/cHbc9+Fe753Rz2VMoEvdsKR+b4h2s7qg2OsIORvfbijJ9bD63/SE3gyO99cIxUQEP9AK78WW+PeAes6wRg63pMboxiLP9wrx0yQhAno/WhQdlqW2zZzTYWn7oyPm0G3aH32pE4dBByWomdt8PyvKtujXD9NCMzBk4+9qVw5AWx7rvYLBVAwtCmPhRWe5G/sMrB99HpL7Qyl26HtUetpmoNTz8ZIXKqUyjhY0pqrCiWzWApfjVnVvgjBNEscULX/zP7rBUbGzyze4WoZRKbFMw6qEvbtrQM7DApeoC+tqe47v88/6CfG4LHn/m0iShg2AVjmYJhHsb9lHR7HTUq2Kf8ZEpUCQ88YEXXoV9ACyd+gF3al0dQdSWdh2kvNhbu8HS/kBSKzJub7yHc/MNy0wsv5ayOAkBByk8cmrADxhAvGLzmFLuo7WPoOC18Nk5dAREdPfx18CCHBo/Suh20Ejqoxg9EymWjo/JiP1GyWAROmOP66CEY7Xj/6rmNF2Dzf6+9+3E8VH6fvMp4R38Dzf4Bgdxj2K8avi82+W4epSszbCN46NczDtHON5caMmgxgU/CEmp59uvbPDG2DSEKLq0yinxMOu+mEkVPdBhIFCrFFECjKNTRoJiCYFB7OQQGRUnEQRhscBclaoxKihYVdAAszAAwCwNACsAVqgHSCdQnyKL6lTrENGimZ9PSoJLGMLgMBivVBVoTYw/D1w5SsiiklIpEQJjw+CUEMMkGmCYgAAL3hWkeGVk7Z5WAUzWMYxoNUu2uudXCTk1Da0gsE4WYCykmzb+H1Y8BIwhxZLTwQxsZBzC6kRlmGS1cOvSzatPH79+AwKmTAlhJ7+EA5M2y59k0AnQcABYJXTEaujvfPXrAfQIFuP/wLp9+X1nCd0m0J+7B7Q7AkCvRBXZTEnSOA406LncEsh60Y5XOKIJHEhS0DAJrW8JSueQAjezu8vJHo9bxQNVp8MjglLMlCyBsv28pMxuraiyRZvVwMEx65D2F3PfH9gkv2nb+ZtIR5IwdFQtqSkQ0lzJqWkrnhnmahaUBAf5CZXlR25nh3yem41ttNuKe68lLmRB+n16pF99bKheUum+7EtkizkHFtW6epKbjYVD/hEjpHDEsxsNlEHCTPhc0GjmTNXgefYr9Pg6wnOVBhDfMtORC6pBPbOw9Y8gkONDzsyIZCqxGD9XDBMI7MXO7cItTCkZIJr1zEkplglkHEkR8jGy+Sgo0G5BsOCbDkX2THQUPYZQxaFJdkov6nqrmC9AZGTv55lmwpurCgZo9fpV8XbytPLY+Q8c6JrJMrX52oZK5mKem6DeaVemBb8BKx53P/P4CwTfx6pzTL6FMWfdSllfygzMwX9iJrw9v7aAeMMjWw6i0n6B7+fHG5SsF03YOONYkeBUyjElrTmVDXltN9FHEyjGsZmyAdt6pNeZw8v/6EnGgVMBHLK+6ErOkwvOZxY8FrUell7ggKq56upaAs17FXxCuYdgjUdRH/O6ytb+YpIemzk+w39I3vbWUY1TQ8gsn85lgTwhKypsbCPFmGpK+WsVXE9uXm4IPjpwQBzEPOv7YYXqhTeY0+sBzZngJ1eDN15uFAP7GvfrEl/P0hxTRwDfqmmNNyE4w43dLu6Mdt/E15sfjTwL8F+WPus31q7BmYT9mBgU5M8x8IDJU07oCCuWzAgstqqInWmH9zk9bnSt1SKldDayd8QE/A6JOdFIX3UB37BbyWLcxLfBWpnq72W2c7grg+PoBvfAY2WOME01GjsdHUtdXQs9PVd9fWcDAxNDQ1tzc427u7YXL4Zevdr7rAfvi17INzm87/KQXwYxfhvC+mOYwF8jlP4bO82ggc9vZx1dPX0DQ3N3L1591isnb9CQYSPGsKOTY6Uflg/Odg0hahCtEEOxNykclzROLuVxr1d+aBqM+dqEqH1dVKPAl7bfO5QsxxjGtDZ/69GedFzBNjmJSwvZ0QbN5pEw7enBFxtmhRhUOluE0CKr7Wm7yjoHJIeauLGbGsIcod1GvmdBKAMlFSTjx4L70ar/FikCGMajKNa4cYxpuTmYFGzw/MFQ027+wWfWqI2CHg0xeW7tigsxp050qECYeOz8TTes7QLVzRvFddD/d/vdg/X+jd3r4H74eetBIDgWWIZlvQV4pSTqLRVf2jdDQHFb9mbrv7DjIx/zGMg0BqOEAKMKoxwgIGyXEXG1dK22sXkzVkBc26g3VgmUSkI3JsECCzEgmS1iJg9XNkGqMGqGLAaGbk7ncjyoV2oYOBjuTZAr3NECAJ4Hi4C7AHDoJIAvlM0dKzjn/+Og81esswHeBuSPugyYTQAe77zTMYAMJ8AA03F+ATZT4VQro2ITMuqAWz0UV71Yx+r9+nD2tmq8SqmqV6lVelVYlVHtUj04R914xZwjc6tOIYUH38opaWaTcapqVa3J0A837b9Gzk7ev/y8Xddk9mJ7tt1mi2Hy93AYCP2hJ3SG9aE0+Pq3j38L+3/49atfD2Ngm/RBlnJlRMcN6F7b9Ppg0K8OqTNa52PGiiVSmfznCn2e06jWaHV6g9FkBoAwi9Vmdzhdbo/X5w8EQ+FINBZvSiRT6QygbgB2+zjEDTbDy7OrG2Eu39MLwPmLQ8MAHD8Fd98Ie/d3ADgAEEzLKID7AX4HHgXDroMR7wI3ArfD6oV511qAwT7LMUDZpAy7TiAKg8O2WlNQApvh+8UbGiKtbBaKSLldsOazZTCUCU7LU/v9WnNQ67BVy8rLbUpOQqmuXFhb7a8tlWorHdbychFVlJcrqFuiVIbNtaYGmVpfI66jJlPpp11JqbY0KBGP+6FWrBZLHUaJWBmw1xqlpaVlCMozRKLyZ4elDRL3kJAUdCCJKVKl3brCAhhrhtjyQRmeal2zta4Qu7jeS8wgNDHWZZsnGYS6udMCcdYNDETzhOH3dShz1xDbyGsxiAZAvGx18XHhAU6oQDQnOo1N/AKQIsGYcJxN1ZiuQrB7BxUgRpdLdMVNBA1TGAGWIujGb0GBSU0HI6exU0eSh0KbQ50ZMmbK8/vAvtZihfazM9Gyg9nyPJTC5ecxUu/ipE8KNlb8AZUndNtIMTS7YykwN7n0UELk9NAM9B5AdYJZU634I0A8jKoNnyvAwkOCMLHk8hJa1MtIj4MQE6NjQTYa8TK1dw4twHlDPYw30ZG1Fack1FnSb1RFCA4f+Ljp9GVbpZBQMzMCP0Xl4CV8TxQUYvk5QBbY0QnG2P8+CFqJGEMiL7UG8Sz2oAqnTKY5lQP5WLp0vIC5Y3DsPY3zthgLiGl4UxTFFrO5S+aXp4uS3oku0pGRVDy+nRGfmYLchVspGQgh+qv6UiNdEZUKetTdSNQ0M9DARKpm7+g1NJWOBg2gsavjowz6fhoeGpsag5pCVv1UkM33KEAg+Dwis4bCVtT9ArF+H/GfnQXUdPE/C0xMLmLLJS4AhmDbqrj0RmAvxMyhLz9MC9johYnLxHN5Xlvl6lEAwxmKnb7Kf5JpbJYpi5eTyRYaH+f8jD3OJgYLgUNG5qdb7OWekY76WXg8zPEnTA4fSpPBycm0NkTlbDY8PZ3REqwx8nCim+jRek1lcDMtYTnVmlMd2kLXRqrc1IMQTMZMJ+yXkol+XA3OsMzt1ha0gZVUMG8JCpobFYemoAjNJcQWsuYzbyrE1meMMhEqp2TX9B/w7lyA51fxsMc1eSsHyrvmgEpyZn3qOtw996Qyf/2dpV+CFUYFXa+b29bNyfgqgCBS1rNLxKW+arpbBI62M4sdndfkrbxQswQlyBQCafAB838DilLxZpNeFK8zPKXD2ETDovlDaHeGyOeaZS2AXPU2rzvaZlaAcRoduWvF6t5ZyFhfJlVmIXexSLm9xAgvyta4rE5rXY3UWKXGwwzEUwY8pQgDTSmtlCKp65PiicbnBFLx8HqWalTCj+BSp8TFzXUGI8vvIvIwbRkYKSvzoHQUjBkEqtngJIi4LrhkVkAlqzSzepq74vzZ3XltoFSUdUybStfLPZqWlh0tYDzfmz3de4UdLWvp2yZXiPo8PX94/RiiHQcYA6n2KJG/JAfKgJqPYcwqWozYpa3E7zzNzw0hMGzWzcqUw83gSVho5E2f2YVBZpvfh1hmkGRsoOhiY1UxsNREE32ZPhwmKKMFpjhoKsPMJow4PdHjFC6TOVm0uqdgVD3e2hFkxeHeeC0dDYwkKUwKySysJZgXH+OaWY9hKoybGKu9PkYoHxKJy9yyqM5jQKi1Ec19/6GrzxKeEqGWvPYfGSF+J2RJYOmF/4RxijFuQovqmeBuBKIMDdGR0ExmRPcdrwI+cergtOaC5tub4feA+bMHtKFJrJWhg8OUxcPP5EDBOA5oF4CU6b/ysGwQhglxQpI/p+www49TZHYdoTlJ1itnGU5czT5zg7GBfn4IHJH8mWOTxYFoRlFTUM8ksrBdz9xuEyyYeNtU2RxxVbtdX6ghp9UeN/HsdT7JZ1vGh10jLWPZZpsx4hpmEzETN4Rj1njYYGiPW2Pt0ECpndIROEXVqhTnQLkzvCkeU9ret6UfwUZpMFozYElIn65JWMWON9Xapxhpll1DLp5PNiXaevrd9t0vvremK/g02fEicIvm0GgXz8h4j+u1mUCDL+hWUWKltFF7OFDnQirdFwL6nT6fzKebmjr6U9rOP7MOQ1vLRPsYfLzooTcp5QWIufFbPXJS00VHDwk+o8RC6edGQhyUGA9RCteyXq1o5/k0ZyO04lmjIZ5wmL+kNdX5W9K7eEma0tdc+EvhitLXAnJxJDK7ttCyBk9qywxO2iezI62RVDpOiY3SBnG9yZe0J+RGkybbJ8rerZEprov/z/m1Qe9YYVvzdhjgpNx16a0BSh2UXkxpFaE8Edy1oUPwiCufTOuH0ydqmgPd6YF4PDXQHVA2f5XXDaef3IqYCxFJ1pAwmQ2JrCRSMKdUvFfsOdWoPuURe6XwsGRTb5ZQGyGGTzVmb8qUrDGYokOi7PNqufDa+P/srQ36puf3ZQptremxEftk8whwmJ65iUiWoLM2KQ4gae+js6WZFJsSZuZ6H+FR9a0vX/0xxCjdpHQ619nd51O3admHrH3J0b6J3g2fuzQZ/84VjLnM94dCl9vtcExnz8ycaK5npjPQ4q2bxik576EKvt3Y6bLak2241Z6yevV9XLPC/ab8QkdR/HCQ2cXZM86bu2rqssLHpvb4yp6VxeRkux6ipJuLP8fU5jPu/fJ6w5jd1/96LxkJYVQbft4i44xs1VEET2tNT04rpnrmOnp61vE0wV4iR2lbh81uSzaLWjYdH/41QsE8plaKL8L0VWuVnmITwbUiyvzmEujAghlPhuRxSRTdJsQSd1vL4xaKt8LbFw2kFx4xfczJEvzGfc4mir2IxrPS/huZ3qLjUiHhmVde+nn1PQGOCEfP/d+tg+00hqQ7fKG1Y+L6teV2Qm2UgBtRfIiSboImz5s6DyVNMCWHjBgbwR112Qpm8v82h/sQ4TAazW1+OMk2NNdxd6IxjXKYXz19f0WiPNHGChqYCskBgl2YnEZxjpcs7fxSOnuYLYUsZOtWKYJjOnrHJgxTbosdHX1jU4aJ3rXHbT+m6tM/2qw/putTP558bzE90D29NNOz0G39N4pRRO+1J7Od9oTTDKInao9Ih6CTm2PVPMUXy0rRVqKPWuhM8p8DD4l/UGzkO4dGO8yDk8xCg5IGeN20LC2hGA373Q8nkl+7PHA3/yl5zkuRm9C1iyyGDbd134nepsv0vSHIZrub2jNTOFeCPVg4NKZXjQiNzxzrir6p+n9zZ3YQLoy2LKNLxcFy/cstlMBv9LQWFl8hJzdGjrYbM3aHMd1xdDyjX1nsKMDlpJunUezHfM/DOWE2xaZKsrmehxAKYLq5KePhdB6tbFLkwyh+RWi1TcomaKrXrZecHm6/Ljh5d5IkG1lZZjUWvQIhP6ZLmwilMTKSroM2NnZ9l4MK3CxgYj49u9dlxnFe7LnqsP4Ac1B3g+eIhI/Q9j2ezeY2GuclodtuNB5kDhpvDN0q5WEr+s6Zd5hJBPHRzw5+v8Yc+u+16KdSPkSUZ97+LkVQibCA1z32t+vc17EqnXif1TlSjNnmjbHPf7GbnstOq8SnFXcC+HgSkV/s/V+Tva52Uk2JVVjSHkxGYqb2X4zyMoqu5Q3mdCwTyaWRz/emptHe6BOJzE+ajh+FNeqXNVcF74Np5q8nytUfx99Sx2Syr6bbbz/WYD3c9OfxcCfcK93U239fvqk3bdp+i7WUODitQPZhg+bn0N2Bk5qGZxQlxEFoW7gNTuMtLyc/bwm2eupepvF362rfd76n7ZRJHzvL/ESmn0fVEmN5cyAVJWazpEyIhevzw/fcOTqa4onu7jmCn46Fv9DUP1qpIE5M6qPpg/pDNs8fGlUfuaET01ushNgosbam7AmXTH/ObmLm5lJHdpdXd+tbuvee07v7Fkrg1teajEYt4+q7R/9cr1MZrVqDZuvzJJ/aVCvz7zx90L+vqlLiSmHtqWQmk2yejgW6hob8XfG4L/K26w9VXeFwl6phwo+7ibVP5HUopaPIQqo7he3wz2ihMS6TvHJZ8+vnNJpuiC9T/Gv0OmA9WeBFo4+AXdzYyDL3HXyraY+Gj9BQ4f6izkzCvDT/jSF7M0Xg282M/VIwc3VCbOdJLS69ubGhZqp4iWiiPd+3h+kKrdcpxVXj7etVUBphYfkthKhNFb4dFGt9d+6R5O50os8/kptRqbjmJ9+D/pMazUn/g77/g2uKBnMpKcqUN/v+qrFKqr8ptj1xvtp8bdN7fe6eo49jqtdCJQXdSCFhWzwfyW8vogOvScsa/+w+obbIZPxU2xNPN1gvjH842NQS0HzkOUyRa2nioD53yqn4459npbTXqeNFZZWoN3krlp6CLbFAB9dKsAfhipZ6T0rJNiU937xo8R/2B4PpUYLdbBLX9SntXkWjKuGpOdeaCKfgdp6X2gyW4LgRG96PsQmRA6PxMJCjOpOrumz3kR3ETmde/zcGe3vTshdvbkmLxcpv+oa2XsoHJ7tDXfhNusnLZo4rcBAjp3yKJppYB01OeWQIBbF5+u+IwC4sH/g7RT5MbOIpnHYIemiyp/7eylmKPAiZyxIo6RAkcKbHJqrmfZgO/h2TNJaK93bjlIWLkFSPaDfCIOL04oWTk5+KpZyfSDUzO36cY3b+OKud1SI/JxVPfbz0q0DGBZHWVJwv28nM/f9Oc1FCYPRZrDQ/sXHChIMYOX7cJmxOsInS7LTtB4T9Ap/kyMUIgwJjlmrOjZxLdjOV9aJBTGwUf4kRi3XnRs9R7GJENZXTmNooBlMJwpXZK/5SUSFOVopF4opSX7tNyoHtJ0K1lExQqqWUUD/ai7M2Iaa/VaMo7+Oif6ey4H/PC+npuoL5Z2t+qjEpbU7tT0pTer1wgJcWZERWkMoG3qUHEoTx6V3DnnFO13GTWCZ6rePdnFbSdkvbTWYChnFiTvjCw0k4Nz+hk4mnFQddNUS5vj6XlSErNa6Ozo/rMPwg3TW+UrbaNZZuS83AmFBoo0LakLHqHujlex/QWR9m3p82oRBGU7PwUaQf45cxzcvR8xeImSjHbxFFHmMQTCAkSCJkRrznUdejUs6CEJtEvB6hv/gfEXDi0KN/RfgopuImRBsoevy2n5rEFMFBTHXnnzeQ/lsm89uj+bTO7e//m6fQkOoohgOXOqgbLnj/1t8fm87zS/xVHKzcMFr+bG/OSUItpjv4yiXPmZ675Cb/xXFDJKlJRgx9Khx4VlccfT5e53QzKaMcs62BiKqy9l3u9Hysia/MPSC4V9kRfIKDj2SDmwi+1Gn7dY2kPlEbvJRkQS5VrqHrPGpcf6Wwi/NKzj+6p93uH4Eq9o2dHRFqvt6oeaQtnFbRtQX1FqnacH1Pq2WmC+Mle2tDHmAy8a4cOu5I+pIxsMVQN8exrf4rR4COVWpiq5zIa3WWgY31CbNZrxlY66RafXIrPHwhFd+ZlR6qSa0b+9raJYsbbM+utdxQ1UJoAaP7sNYbqqeZtkofEF2/ZTeurwjGDgIe1PIx4K2LCN1obDkYGT3i1PFP/R0LqyA3ApiMvf971dbFyzFAd8pP/xs5fvfqqOi/pKQz8GnMB7Pd35JzqR4DBmMBAt4YbBkYMg9Wx7hIcw/MDZLDQAd4hhtfT6fcwZBasnTQpk9j+uHspMg8QxLkmVKbBdqMNJtV3lzAkkujpQtBynI1Z25IcW+5jfbfKfLG7vmKNTcK7jKWDIFd43OAIhmiXMGZdcIaqSynGoZCOUSY7RqTY0+WHvTGtAykX4jSTJ4xdjcBb4JsGXrsJWBu88KSW6e99wajf/wS6NT7P6Lk2RNXlT+RxE+YHSqyQJVNkLmgMwC8oN77wHfAD0AH5HP6KqV3PbG7gY9AJhCnEWQO8CVL/kDnOLXjVObbmwDIB2z5D/UdnZcrM6c5nJ1fGyGwHSyCKdBim9wM+sX8EPIGiom29+vCEeQiisIAox3bnHXO40MulElyBedursOYY1yXunxcT1k9zPU11Gu9H6mmoyTAdIcAn4CCY3hTwrHKc4kT6M0ujpPJSY6nrGEOiddZ46ecvz6XVrTNfjssW7RkFxU3J5cQlf4GKHjnFS1YNlMOzIwNO6n02qFoRcHcNGfS7thySdEOOjXCOXfZxrzDHBwWLXPc/m6z7OYUrXNYU+CjhWmW4jByG4dKrS3bBL4jCfg1ufx/FsPcfHFhecanz2x8W2nCt5BwOOMvhiJpcbmSvnv2Pyu0SXH3NsdEla0iJrBS2myfXXZsNKQV3fMKtNMWzAPV6Ks0ZqWLG78UGksBe3dCwaLd1szYAb6ZUynAenYujD0RnfKeyCYd6WR0hdqAvdevdDldkc58xBjYoIY0NpvsAnCxtIq8oF12qCrbmcWky+JCVbUNvnAVMrpYwfUesjlwDPKRzkW7ZZtw72Q30b3GHrW3tMihR1YnCBzh963g3rm89nVCpD3ZToBy0dC2w+3RfD8QbIpAhUpiElIyctUUlGrU0tLRMzAyMbOwsrFzcHJx8/Dy8QsICgmLiIrJaJbVolWbdh06denWo1efnLx+A+/2+rAZNWbcRATpeie8d33kvSCPe8LTnvGGJz3lTed7wMuOejE4g10cmqFOpiTDMyIjMyqjI0pVxJFEGlnkqY4iykqfwLvvXXY6k857NhPnTqXXdTudKVfAy333K/HxzIM7int3b6PwC4ZgdpcngL8L6w52LaDivYWjlF2PEZR8ZROzEfEXPIH38NuhNo/jdt6xD31hF3iMvkoLPI7buGOD0h3gO70hBbEeJ77XuM8/SXfdsTZf+BknO3w2p+Yn3UsAAA==) format(\'woff2\'); }                .badge { font-size: 38px; font-weight: bold; }                .address { font-family: \'Sofia Sans Extra Condensed\', sans-serif; font-size: 22px; font-weight: 300; text-transform: lowercase; filter: drop-shadow(2px 1px 3px rgb(0,0,0,0.75)); }                .logo { filter: drop-shadow(1px 3px 2px rgb(0,0,0,0.75)); }                .nft-address { filter: drop-shadow(2px 1px 7px rgb(0, 0, 0, 0.5)) }      </style><defs><linearGradient id="grad1" x1="0%" y1="0%" x2="74%" y2="100%"><stop offset="0%" style="stop-color:rgba(86,111,149,1)"></stop><stop offset="48%" style="stop-color:rgba(35,67,116,1)"></stop><stop offset="100%" style="stop-color:rgba(35,67,116,1)"></stop></linearGradient><linearGradient id="badge" x1="0%" y1="0%" x2="74%" y2="100%"><stop offset="0%" style="stop-color:rgba(86,111,149,1)"></stop><stop offset="48%" style="stop-color:rgba(35,67,116,1)"></stop><stop offset="100%" style="stop-color:rgba(35,67,116,1)"></stop></linearGradient></defs><rect x="24" y="24" width="400" height="500" rx="20" fill="white" class="anim"></rect><rect x="34" y="34" width="380" height="480" rx="20" fill="url(#grad1)"></rect><circle cx="40" cy="40" r="40" fill="white"></circle><circle cx="40" cy="40" r="34" fill="url(#badge)"></circle><text x="40" y="44" dominant-baseline="middle" text-anchor="middle" fill="white" class="badge">',
          Strings.toString(registrar.getAddressDifficulty(tokenId)),
          '</text><g class="logo"><svg x="175" y="25%" width="97.35" height="154.8" viewBox="0 0 115 182" fill="none" xmlns="http://www.w3.org/2000/svg"><path d="M57.5054 181V135.84L1.64064 103.171L57.5054 181Z" fill="#88AAF1" stroke="#1616B4" stroke-linejoin="round"></path><path d="M57.6906 181V135.84L113.555 103.171L57.6906 181Z" fill="#b0d5fb" stroke="#1616B4" stroke-linejoin="round"></path><path d="M57.5055 124.615V66.9786L1 92.2811L57.5055 124.615Z" fill="#B8FAF6" stroke="#1616B4" stroke-linejoin="round"></path><path d="M57.6903 124.615V66.9786L114.196 92.2811L57.6903 124.615Z" fill="#b0d5fb" stroke="#1616B4" stroke-linejoin="round"></path><path d="M1.00006 92.2811L57.5054 1V66.9786L1.00006 92.2811Z" fill="#88AAF1" stroke="#1616B4" stroke-linejoin="round"></path><path d="M114.196 92.2811L57.6906 1V66.9786L114.196 92.2811Z" fill="#B8FAF6" stroke="#1616B4" stroke-linejoin="round"></path></svg></g><text x="50%" y="404" transform="translate(12 0)" dominant-baseline="middle" text-anchor="middle" fill="white" class="address">',
          Strings.toHexString(address(uint160(tokenId))),
          "</text></svg></g></svg>"
        )
      );
  }

  /// @notice Returns the metadata for and NFT address
  /// @param tokenId ID of the NFT address
  /// @param registrar Registrar
  function getMetaData(uint256 tokenId, Registrar registrar) public pure returns (string memory) {
    return
      string(
        abi.encodePacked(
          '{ "name": "Lambda Address ',
          Strings.toHexString(address(uint160(tokenId))),
          '", "image": "',
          abi.encodePacked(
            "data:image/svg+xml;base64,",
            Base64.encode(bytes(getImage(tokenId, registrar)))
          ),
          '", "description": "Lambda Address" }'
        )
      );
  }
}

// File: src/contracts/NFTAddressFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

import {Errors} from "./Errors.sol";
import {IFactory} from "./IFactory.sol";
import {Registrar} from "./Registrar.sol";
import {SafeCreate3} from "./SafeCreate3.sol";

/// @title NFTAddressFactory
/// @notice Main Factory used to mint and deploy NFT addresses.
contract NFTAddressFactory is IFactory {
  Registrar private _registrar;

  mapping(address nftAddress => uint256 salt) private salts;

  /// @param registrar Address of the Registrar
  constructor(Registrar registrar) {
    _registrar = registrar;
  }

  /// @notice Deploys a smart contract to an NFT address
  /// @param nftAddress NFT address
  /// @param creationCode Smart contract bytecode (i.e. bytecode containing the constructor and
  /// initialization data)
  function deploy(address nftAddress, bytes memory creationCode) external {
    require(msg.sender == address(_registrar), Errors.NOT_REGISTRAR);

    address mintedBy = _registrar.getMintedBy(uint256(uint160(nftAddress)));
    address deployedAddress = SafeCreate3.deploy(
      bytes32(salts[nftAddress]),
      mintedBy,
      creationCode,
      0
    );

    require(deployedAddress == nftAddress, Errors.INVALID_ADDRESS_CREATED);
  }

  /// @notice Returns the NFT address created by `mintedBy` with `salt`
  /// @param mintedBy Account that minted the NFT address
  /// @param salt Salt used to generate the NFT address
  function getAddress(address mintedBy, uint256 salt) public view returns (address payable) {
    return payable(SafeCreate3.getDeployed(bytes32(salt), mintedBy));
  }

  /// @notice Mints a new NFT address, created by `mintedBy` with `salt`
  /// @param mintedBy Account that is minting the NFT address
  /// @param salt Salt used to generate the NFT address
  function mint(address mintedBy, uint256 salt) external returns (address payable) {
    require(msg.sender == address(_registrar), Errors.NOT_REGISTRAR);

    address payable addr = getAddress(mintedBy, salt);
    salts[addr] = salt;
    return addr;
  }
}

// File: src/contracts/Registrar.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

import {ERC721Upgradeable, IERC165Upgradeable} from "@openzeppelin/contracts-upgradeable/token/ERC721/ERC721Upgradeable.sol";
import {Errors} from "./Errors.sol";
import {OwnableUpgradeable} from "@openzeppelin/contracts-upgradeable/access/OwnableUpgradeable.sol";
import {IFactory} from "./IFactory.sol";
import {IMetaData} from "./IMetaData.sol";

/// @title Registrar contract
/// @notice Main contract to mint and query NFT addresses. Only addresses for smart contracts
/// can be minted, no EOA. This contract implements the ERC721 standard.
/// @dev This contract is meant to be initialized with a proxy, and every interaction should
/// be through the proxy. Logic is further divided into 2 main contracts:
///   - [IFactory](IFactory.md): Registration, validation and deployment of NFT addresses
///   - [IMetaData](IMetaData.md): Generating the metadata for the NFT addresses.
///
/// The ID of an NFT address is simply its value casted to `uint256`.
contract Registrar is ERC721Upgradeable, OwnableUpgradeable {
  string private constant _NAME = "Lambda Address";
  string private constant _SYMBOL = "LADD";
  bytes4 private constant _INTERFACE_ID_ERC2981 = 0x2a55205a;

  /// @dev Emitted when a contract is deployed to the `tokenId` NFT address.
  event Deploy(address indexed deployer, uint256 indexed tokenId);

  /// @dev Emitted when `owner` enables `approved` to deploy to the `tokenId` NFT address.
  event DeployerApproval(address indexed owner, address indexed approved, uint256 indexed tokenId);

  /// @dev Represents an NFT address registration
  struct NFTAddress {
    address mintedBy; // Account that minted the NFT address
    bool isDeployed; // Is a smart contract deployed at the address?
    IFactory factory; // Factory used to mint and deploy the address
  }

  mapping(uint256 tokenId => NFTAddress) private _nftAddress;
  mapping(IFactory => bool isAllowed) private _allowedFactories;

  uint256 private _mintPrice; // Mint price in wei
  IMetaData private _metaData; // Contract used to generate metadata for the NFT addresses

  // _royalties and _royaltiesRecipient are not currently used
  uint96 _royalties;
  address payable _royaltiesRecipient;

  // Account that can deploy to the NFT address, other than its owner
  mapping(uint256 tokenId => address account) private _deployApprovals;

  /// @notice Initializes the registrar
  /// @dev Function is invoked by the proxy contract during main deployment
  /// @param mintPrice Price in wei to mint a new NFT address
  /// @param metaData IMetaData contract to handle metadata generation
  /// @param owner Contract's owner
  function initialize(uint256 mintPrice, IMetaData metaData, address owner) external initializer {
    __ERC721_init(_NAME, _SYMBOL);
    _mintPrice = mintPrice;
    _metaData = metaData;

    OwnableUpgradeable._transferOwnership(owner);
  }

  /// @notice Allow/disallow a factory
  /// @param factory IFactory contract
  /// @param allow Allow/disallow the factory
  function allowFactory(IFactory factory, bool allow) external onlyOwner {
    _allowedFactories[factory] = allow;
  }

  /// Approve `deployer` to deploy to the `tokenId` NFT address.
  /// @param deployer Account that can deploy a smart contract
  /// @param tokenId ID of the NFT address
  function approveDeployer(address deployer, uint256 tokenId) external {
    require(msg.sender == ERC721Upgradeable.ownerOf(tokenId), Errors.ERC721_NOT_TOKEN_OWNER);
    _approveDeployer(deployer, tokenId);
  }

  /// @notice Returns `true` if factory is allowed to mint new NFT addresses
  /// @param factory Factory
  function isFactoryAllowed(IFactory factory) external view returns (bool) {
    return _allowedFactories[factory];
  }

  /// @notice Deploys a smart contract to the NFT address
  /// @param tokenId ID of the NFT address
  /// @param creationCode Creation bytecode of the smart contract
  function deploy(uint256 tokenId, bytes memory creationCode) external {
    require(_canDeploy(tokenId, msg.sender), Errors.NO_DEPLOY_PERMISSION);
    address nftAddress = address(uint160(tokenId));
    _nftAddress[tokenId].isDeployed = true;
    getFactory(tokenId).deploy(nftAddress, creationCode);
    emit Deploy(msg.sender, tokenId);
  }

  /// @notice Returns the ID for an NFT address.
  /// @dev The ID is the address' value casted to uint256. Hence there
  /// can't be any ID larger than 20 bytes.
  /// @param nftAddress NFT address
  function getAddressId(address nftAddress) external pure returns (uint256) {
    return uint256(uint160(nftAddress));
  }

  /// @notice Returns the number of consecutive identical characters in an NFT address.
  /// The higher the number, the greater the difficulty to generate the address.
  /// `e.g.: 0x6e495383a99fde33333337cae062d75153fa2608 -> 7`
  /// @param tokenId ID of the NFT address
  function getAddressDifficulty(uint256 tokenId) external pure returns (uint8) {
    uint8 highestHex = 0;
    uint8 highestHexCount = 0;
    uint8 currentHex = 0;
    uint8 currentCount = 0;

    uint8[] memory it = new uint8[](2);

    for (uint8 i = 0; i < 20; ) {
      uint8 b = uint8(uint(uint160(tokenId)) / (2 ** (8 * (19 - i))));
      it[0] = uint8(b) / 16;
      it[1] = uint8(b) - 16 * it[0];

      for (uint8 j = 0; j < 2; ) {
        if (it[j] == currentHex) {
          unchecked {
            currentCount++;
          }
        } else {
          currentHex = it[j];
          currentCount = 1;
        }

        if (currentCount > highestHexCount) {
          if (highestHex != currentHex) highestHex = currentHex;

          highestHexCount = currentCount;
        }

        unchecked {
          j++;
        }
      }

      unchecked {
        i++;
      }
    }
    return highestHexCount;
  }

  /// @notice Returns the approved deployer for a given NFT address
  /// @param tokenId ID of the NFT address
  function getApprovedDeployer(uint256 tokenId) external view returns (address) {
    return _deployApprovals[tokenId];
  }

  /// @notice Returns the Factory used to mint and deploy an NFT address
  /// @param tokenId ID of the NFT address
  function getFactory(uint256 tokenId) public view returns (IFactory) {
    return _nftAddress[tokenId].factory;
  }

  /// @notice Returns the address of the account that minted the NFT address
  /// @param tokenId ID of the NFT address
  function getMintedBy(uint256 tokenId) external view returns (address) {
    return _nftAddress[tokenId].mintedBy;
  }

  /// @notice Returns `true` if a smart contract has been deployed to the NFT address
  /// @param tokenId ID of the NFT address
  function getIsDeployed(uint256 tokenId) external view returns (bool) {
    return _nftAddress[tokenId].isDeployed;
  }

  /// @notice Mint a new NFT address. This function doesn't deploy a smart contract
  /// to the address.
  /// @param factory Factory used to mint and deploy the NFT address
  /// @param salt  Salt used to generate the NFT address
  function mint(IFactory factory, uint256 salt) external payable returns (uint256 tokenId) {
    require(_allowedFactories[factory], Errors.UNAUTHORIZED_FACTORY);
    require(msg.value >= _mintPrice, Errors.INSUFFICIENT_MINT_PAYMENT);

    address payable deployedAddress = factory.mint(msg.sender, salt);
    _mint(deployedAddress, msg.sender, factory);
    return uint256(uint160(address(deployedAddress)));
  }

  /// @notice Set the contract used to generate metadata for the NFT addresses
  /// @param metaData IMetaData contract address
  function setMetaData(IMetaData metaData) external onlyOwner {
    _metaData = metaData;
  }

  /// @notice Set the price of minting NFT addresses, in wei.
  /// @param mintPrice Mint price in wei
  function setMintPrice(uint256 mintPrice) external onlyOwner {
    _mintPrice = mintPrice;
  }

  /// @notice Returns an URI for a given token ID. The URI is generated
  /// by [IMetaData](IMetaData.md).
  /// @param tokenId ID of the NFT address
  function tokenURI(uint256 tokenId) public view override returns (string memory) {
    return _metaData.getTokenURI(tokenId, this);
  }

  /// @notice Withdraw `amount` of wei from this contract and transfer it to recipient `to`.
  /// @param amount Amount of wei to withdraw
  /// @param to Address of the recipient
  function transferFunds(uint256 amount, address payable to) external onlyOwner {
    require(amount <= address(this).balance, Errors.INSUFFICIENT_FUNDS);
    (bool succeeded, ) = to.call{value: amount}("");
    require(succeeded, Errors.TRANSFER_FAILED);
  }

  function _approveDeployer(address deployer, uint256 tokenId) internal {
    _deployApprovals[tokenId] = deployer;
    emit DeployerApproval(ERC721Upgradeable.ownerOf(tokenId), deployer, tokenId);
  }

  function _beforeTokenTransfer(address, address, uint256 tokenId) internal virtual override {
    if (_deployApprovals[tokenId] != address(0)) {
      _approveDeployer(address(0), tokenId);
    }
  }

  function _canDeploy(uint256 tokenId, address deployer) internal view returns (bool) {
    return deployer == ERC721Upgradeable.ownerOf(tokenId) || deployer == _deployApprovals[tokenId];
  }

  function _mint(address payable addr, address mintedBy, IFactory factory) internal {
    uint256 tokenId = uint256(uint160(address(addr)));
    require(!_exists(tokenId), Errors.ADDRESS_ALREADY_MINTED);

    _mint(mintedBy, tokenId);
    _nftAddress[tokenId].mintedBy = mintedBy;
    _nftAddress[tokenId].factory = factory;
  }
}

// File: src/contracts/RegistrarProxy.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

import {Errors} from "./Errors.sol";
import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";
import {ERC1967Proxy} from "@openzeppelin/contracts/proxy/ERC1967/ERC1967Proxy.sol";

/// @title RegistrarProxy
/// @notice Proxy used to upgrade and delegate logic to Registrar. All interactions with
/// the Registrar should be made through this proxy.
contract RegistrarProxy is ERC1967Proxy {
  /// @param implementation Address of the Registrar
  /// @param data Initialization call data
  constructor(address implementation, bytes memory data) ERC1967Proxy(implementation, data) {}

  /// @notice Upgrades the implementation logic of the proxy.
  /// @param newImplementation Address of the new Registrar
  function upgradeTo(address newImplementation) external {
    require(Ownable(address(this)).owner() == msg.sender, Errors.OWNABLE_NOT_OWNER);
    _upgradeTo(newImplementation);
  }
}

// File: src/contracts/SafeCreate3.sol



// SPDX-License-Identifier: AGPL-3.0-only
pragma solidity 0.8.18;

import {Bytes32AddressLib} from "solmate/src/utils/Bytes32AddressLib.sol";
import {CREATE3} from "solmate/src/utils/CREATE3.sol";

/// @title SafeCreate3
/// @notice Variant of solmate's CREATE3 with front running protection
/// @dev Since `deploy(...)` is always called from the same address (i.e. the Factory),
/// NFT address registration could technically be front run. To mitigate this, we add
/// at the begining of the bytecode the address of the account that creates it.
/// e.g.: PUSH20 `createdBy`. This operation doesn't do anything other than prevent
/// front running. The rest of the library is practically identical to solmate's CREATE3.
library SafeCreate3 {
  using Bytes32AddressLib for bytes32;

  function deploy(
    bytes32 salt,
    address createdBy,
    bytes memory creationCode,
    uint256 value
  ) internal returns (address deployed) {
    bytes memory proxyChildBytecode = getBytecode(createdBy);

    address proxy;
    /// @solidity memory-safe-assembly
    assembly {
      // Deploy a new contract with our pre-made bytecode via CREATE2.
      // We start 32 bytes into the code to avoid copying the byte length.
      proxy := create2(0, add(proxyChildBytecode, 32), mload(proxyChildBytecode), salt)
    }
    require(proxy != address(0), "DEPLOYMENT_FAILED");

    deployed = getDeployed(salt, createdBy);
    (bool success, ) = proxy.call{value: value}(creationCode);
    require(success && deployed.code.length != 0, "INITIALIZATION_FAILED");
  }

  function getDeployed(bytes32 salt, address createdBy) internal view returns (address) {
    address proxy = keccak256(
      abi.encodePacked(
        // Prefix:
        bytes1(0xFF),
        // Creator:
        address(this),
        // Salt:
        salt,
        // Bytecode hash:
        getBytecodeHash(createdBy)
      )
    ).fromLast20Bytes();

    return
      keccak256(
        abi.encodePacked(
          // 0xd6 = 0xc0 (short RLP prefix) + 0x16 (length of: 0x94 ++ proxy ++ 0x01)
          // 0x94 = 0x80 + 0x14 (0x14 = the length of an address, 20 bytes, in hex)
          hex"d6_94",
          proxy,
          hex"01" // Nonce of the proxy contract (1)
        )
      ).fromLast20Bytes();
  }

  //--------------------------------------------------------------------------------//
  // Opcode     | Opcode + Arguments    | Description      | Stack View             //
  //--------------------------------------------------------------------------------//
  // 0x73       |  0x73 createdBy       | PUSH20           | createdBy              //
  // 0x50       |  0x50                 | POP              |                        //
  // ... See CREATE3's bytecode
  //
  /// @notice Adds "PUSH20 `createdBy`" and POP to solmate's CREATE3 deploy bytecode
  /// to prevent frontrunning.
  /// @param createdBy Address of the account that creates the smart contract
  function getBytecode(address createdBy) internal pure returns (bytes memory) {
    return abi.encodePacked(hex"73", createdBy, hex"50", CREATE3.PROXY_BYTECODE);
  }

  function getBytecodeHash(address createdBy) internal pure returns (bytes32) {
    return keccak256(getBytecode(createdBy));
  }
}